
In this first part of a two-part article, we introduce Clipper 5.0's powerful new array handling capabilities. We start by showing 

how to create arrays, continue by showing how to change their size dynamically, and finish by looking at their implementation. 

 

The old-style arrays In Clipper Summer '87, you create an array in one of three ways. You use either the DECLARE, PRIVATE, 
or PUBLIC declarative statements, followed by a numeric expression enclosed in square brackets, as in: 

 
PUBLIC inv_dates[ 10 ]  
PRIVATE field_names[ fcount() ] 
 

DECLARE is redundant and should be avoided. It's a synonym for PRIVATE. The array size is fixed when you declare it you can't 

increase or decrease it. 

 

Summer '87 arrays are vectors--they have one dimension. One of the most requested features for the new Clipper was multi-

dimensional arrays. As you'll see in Part 2, Nantucket has implemented this (and more). 

 

Declaring arrays in Clipper 5.0 

 

You can still create PUBLIC and PRIVATE arrays in 5.0--but you can also create LOCAL and both INTERNAL and EXTERNAL 
STATIC arrays, usually a better choice. Here are some examples: 

 
// Compile this with /n 
 
#define MAX_DEPTH 10  
STATIC stack[ MAX_DEPTH ]  // A "file wide" (extern al) STATIC array of 10 elements  
  
FUNCTION maintain_sets ( set_num, new_val ) 
#define MAX_SETS 10        // Internal static array also of 10 elements  
STATIC sets[ MAX SETS ]    // Can only be seen insi de this function   
LOCAL old_val 
 
IF set_num > 0 .AND. set_num <= MAX SETS  
   old_val = sets[ set_num] 
   IF new_val != NIL  
      sets[ set_num] = new_val  
      . .  
   ENDIF  
ENDIF 
RETURN old_val 
 
FUNCTION proc_fields 
LOCAL num_fields := fcount() 
LOCAL fnames[ num_fields ]   // A local array size the same as the number of  
                             // fields in the curre ntly selected database  
afields ( fnames )  
FOR i = 1 TO NUM_FIELDS  
      . .  
NEXT 
 
RETURN NIL 
 

LOCAL and internal STATIC declarations must appear inside a function, but before any executable code. If you don't know 

the array's size at this time, however, you just declare the name. Later, when you do know its size, you create it using 

the array function, as in: 
 
FUNCTION test 
LOCAL ar           // valtype(ar) = "U" here  
                   // ar is NIL here  
. .  
ar = array( size ) // valtype (ar) is "A" here  
                   // len(ar) is “size” here 
 
The array function returns an array of the specified size. Each element is "nil." 

 

 

Array literals 

 

You can initialize an array when you declare it using array constants and the in-line assignment operator. You write 

an array constant as a comma-separated list of expressions enclosed in curly braces ({}). The following example shows this: 

 
FUNCTION quarter_name( quarter_num ) 
LOCAL quarters := {"January", "April", "July", "Oct ober"} 
                      // valtype(quarters) = "A"  
                      // len(quarters) = 4 
RETURN quarters[quarter num]  

 

We declare “quarters” as a variable with a LOCAL scope and initialize it to an array with four elements. The declaration 

produces the same result as: 



 
LOCAL quarters[4] 
quarters[1] = "January"  
quarters[2] = "April"  
quarters[3] = "July"  
quarters[4] = "October"  

 

You can use array constants in any expression, so you can rewrite the previous example as: 

 
FUNCTION quarter_name(quarter_num) 
RETURN {"January", "April", "July", "October"}[ qua rter_num ] 
 
A common task is to return an empty value of a certain type. You can write this as: 

 
FUNCTION empty_val ( var_type) 
RETURN {{}, "", ctod(""), .F., 0} [at (upper (var t ype), "ACDLN")]  

 

The literal array: 

 

{{}, "", ctod(""), .F., 0} 

 

contains "empty" values for each data type in ascending order of their type names (Array, Character, Date, Logical, and 

Numeric). We index it by the position of the variable's type in the character string "ACDLN." 

 

Certain functions expect arrays as parameters. You can, of course, pass an array constant instead, as in: 
 
achoice(10, 10, 20, 20, {"Amex", "M/C ", "Visa"}) 
 

Changing an array's size 

 

While it was awkward to change an array's size in Summer '87, in 5.0 it's easy. ASIZE() changes the size of an existing array, 

as in: 

 
ar = {"Choice 1", "Choice 2", "Choice 3", "Choice 4 "}  
? len(ar) // 4 
asize(ar, 7) // ar = {"Choice 1", "Choice 2", "Choi ce 3",  
             //       "Choice 4", NIL, NIL, NIL}  
? len(ar) // 7 
 

The call to ASIZE()  changes the size of “ar” to 7. The first four elements remain the same the three new elements have a "nil" 

value. If you make the array smaller, the other elements are released. The difference between ASIZE()  and ARRAY() is that 

ASIZE()  changes the size of an existing array, whereas ARRAY() creates a new, empty array. 

 

If you simply want to add an element to the end of an array, use the AADD()  function, as in: 

 
// Menu (.dbf) is selected 
FUNCTION build_menu 
LOCAL choices := {}  
DO WHILE !eof() aadd( choices, menu -> choice )  
     SKIP  
ENDDO 
RETURN choices  

 

Or you can use DBEVAL(): 
 
FUNCTION build_menu 
LOCAL choices := {} 
dbeval({|| aadd(choices, menu -> choice)}) 
RETURN choices 
 

AADD() does two jobs. It increases the size of an array by one, then initializes the new element to its second parameter. The 

following two code sections, for example, create the same array: 

 
// Section 1  
choices := {"Choice 1", "Choice 2", "Choice 3"} 
// Section 2  
choices := {}  
aadd (choices, "Choice 1")  
aadd (choices, "Choice 2")  
aadd (choices, "Choice 3")  

 

AADD()  is useful when you don't know the size of the array in advance, such as when you're processing a database with a 

scope. It saves you from executing a COUNT first. 
 

Note how you create an empty array: 

 
LOCAL choices := {} // len (choices) = 0  
                    // valtype (choices) = "A"  

 

This is important,since AADD()  must operate on an array. If you declare choices without initializing it, AADD()  will produce an 

error. It expects an array as a parameter. 

 

 

 



Arrays as functions return results 

 

As we've been implying, functions can return arrays in 5.0. This wasn't possible in Summer '87. Functions should return arrays 

when they need to return more than one value. This is preferable to passing many parameters by reference. 

 

Two new functions that return arrays are DBSTRUCT() and DIRECTORY(). I'll cover them in more detail in a later article, but 

for now, here's a quick look at what they do. 

 

DBSTRUCT() returns a two-dimensional array containing information about the currently selected database. Its format is 

similar to the database created by the COPY STRUCTURE EXTENDED command. You can use DBSTRUCT() instead of AFIELDS(). 
AFIELDS()  expects a number of arrays as parameters and fills them with the structure of the currently selected database, as 

in: 

  
LOCAL field_names [ fcount() ],;  
      field_types [ fcount() ],;  
      field_lens  [ fcount() ],;  
      field_decs  [ fcount() ] 
 
afield (field_names, field_types, field_lens, field _decs) 
 
DBSTRUCT() returns the same information in a two-dimensional array, as in: 
 
LOCAL file_stru := dbstruct() // valtype (file_stru ) = "A"  
                              // len (file_stru) = fcount()  

 

There's one row for each field in the database the first column of each row contains the field name, the second the field type, 

the third the field length, and the fourth the number of decimal places. 

 

DIRECTORY() is a replacement for ADIR(). ADIR()  is used to fill arrays with information about disk files. DIRECTORY() does the 

same job, except it returns a two-dimensional array. 

 

Arrays are implemented referentially 

 
In 5.0, variables of type array are pointers, or references to arrays. An array variable doesn't contain an array, it just refers 

to an array. The array exists in its own right, not just as part of variable. 

  

Now, you can make more than one variable point to an array. For example, writing: 

 
new_choices = choices  
 

doesn't copy the choices array into new_choices. You must use ACOPY() to do that. Rather, it makes new_choices and choices 

point to, or refer to, the same array.  

 

Now, there's only one array, but we have two variables pointing to it. Once consequence is that changing an element makes 

that change visible in all variables that point to that array, as in: 

 
choices = {"Choice 1", "Choice 2", "Choice 3"}  
new_choices = choices 
choices [2] = "Modified"  
? new choices [2]  // Modified 
 

Clipper tracks how many variables point to an array (its "reference count") and only releases the array when it reaches 0. For 

example: 

 
PRIVATE choices := {"Choice 1", "Choice 2", "Choice  3"}  
PRIVATE new_choices := choices  
. .  
RELEASE choices // {"Choice 1", "Choice 2", Choice 3"} array still  
                // exists, its reference count is 1   
. .  
RELEASE new_choices // Clipper releases the array 
 

You can use Clipper's "==" operator to determine whether two variables refer to the same array. "==" returns ".T." if two 

arrays refer to the same array otherwise, it returns ".F." Note that this only checks whether they refer to the same array. you 

can't use "==" to compare two arrays, as in: 

 
ar1 = {1, 3, 5}  
ar2 = ar1  
ar3 = {1, 3, 5}  
? arl == ar2 // .T.  
? ar3 == ar1 // .F.  

 

Even through AR3 and AR1 are arrays of the same size and their elements contain the same values, "==" returns ".F." because 

they refer to different arrays. 

 
  



Arrays as parameters 

 

There's a subtle consequence of this array implementation when you consider how arrays are passed as parameters. In 

Summer '87, entire arrays are passed by reference. This means you can change any of the elements. 

 

The same applies in 5.0 except there's one extra thing to take into account. We have the array variable and the array.  

 

Remember, these are distinct in 5.0. 

 
When you call a function passing an array as a parameter, as in: 

 
ar = {1, 3, 5} 
test (ar) 
 
FUNCTION test(formal_ar)  
. .  
RETURN NIL 
 

you've created two array variables pointing to the same array. 

  

You may or may not be able to access “ar” from inside test that depends on its scope. It it's PRIVATE, PUBLIC, or EXTERNAL 

STATIC in the same PRG file, it's visible. If it's LOCAL, INTERNAL STATIC, or EXTERNAL STATIC  in a different PRG file, it isn't 

visible. 

 

However, whether or not you can access AR, you can still access its elements “formal_ar” points to the same array. If, from 
inside function test, you write: 

 
formal_ar [3] = 7 
 

on return, ar[3] is also 7 they refer to the same array. 

 

You can increase the array's size from inside test, as in: 

 
aadd(formal_ar, 7) 
 

On return, the length of “ar” is 4, and its fourth element contains the value 7. 

 

What this means is that function test receives a pointer to the array, and thus can change anything about the array. One thing 
that may appear confusing crops up while executing the following from inside function test: 

 
formal_ar = {"Rick", "Spence"} 
 

This doesn't affect the “ar” array. It simply creates a new array, points “formal_ar “ to it, and decrements the reference count 

of the {1,3,5} array to 1 (calling the function made it 2). When function “test” returns, the {"Rick", Spence"} array is 

released its reference count becomes 0. 

 

In 5.0 you can pass the "array variable" by reference, as in: 

 
ar = {1, 3, 5}  
test (@ar) 
 

Now, if you assign to “formal_ar “ from inside test, you really do change “ar”, as in: 

 
FUNCTION test (formal_ar) 
   formal_ar = .T. 
RETURN NIL 

 

On return from test, “ar” is the logical constant ".T." You can also assign it another array of course, as in: 

 
FUNCTION test (formal_ar) 
   formal_ar = {"Rick", "Spence"} 
RETURN .T.  

 

So what's the difference between 5.0 and Summer '87 ?  Not that passing arrays to functions has changed, but that you can 

separate the array variable from the array and pass it by reference, thereby allowing it to be changed. For more details, I 

suggest you consult the "Basics Concepts" chapter of the Clipper 5.0 Reference Guide. 

